CALIBRATING THE WAVELENGTH SCALE

In a standard Michelson interferometer, the ordinate axis of the interferogram is path difference measured in cm (equal to twice the slide displacement). Therefore, the transformed spectra will be in units of cm^{-1}, called wavenumbers and denoted by ν. Since the path difference of a field-widened Michelson is a non-linear function of slide displacement, the ordinate axis of the transformed spectra will be a non-linear function of true wavenumber. However, it is convenient to treat the field-widened case as if it were a standard Michelson and thus the spectra will be given as a function of slide wavenumber, σ, and a wavelength calibration must be performed to determine the non-linear relationship between σ and the true wavenumber, ν. For the non field-widened case, slide wavenumbers (Scm^{-1}, or σ) are equal to real wavenumbers (cm^{-1} or ν), but the field-widening causes the path difference to be equal to the slide displacement times a term $\approx(n-1)$, where n is the index of refraction of the optical components doing the field-widening. Thus, the Scm^{-1} scale is equal to the true cm^{-1} scale times a term that looks like (n-1). Therefore, the wavelength calibration is performed by taking lines of known wavenumber, ν, and observing the slide wavenumber, σ, at which they appear in the transformed spectra. The free parameters in the classical expression for (n-1) vs. ν are then adjusted to give the best fit in a least squares sense to σ as a function of ν. The expression used is:

$$\sigma = \nu * (A + B/\nu**2 + C/\nu**4 + D * \nu**2)$$

where A, B, C, and D are the free parameters in the least squares fit.

In order to do the wavelength calibration, one needs to first get the approximate location (in slide wavenumbers) of the peaks of lines of known wavenumber, ν, from either a neon/argon spectrum or an OH spectrum. The easiest way to do this is to read them off from a plot of the spectra. Next, a data file is created with the approximate slide wavenumber, σ, and the corresponding real wavenumber, ν, in two column format and ordered in increasing wavenumber. The programme [PENDLETN.FWI]PEAKFND may then be run. The programme prompts for the calibration spectrum filename, the filename of the approximate peak positions, and the name of the output file that will contain the exact σ of the line, a weighting factor (related to the "sharpness" of the peak), and the corresponding ν. The output of this programme may then be used as input to the programme [PENDLETN]WAVEFIT, which fits the data with the expression given above and outputs the parameters A, B, C, and D as well as a detailed list of the real wavenumber, ν, σ calculated from the fit, σ of the peak in the data, and the difference between the two σ's. These parameters may then be used as coefficients in the subroutines GETSWN, GETRWN, and SWPRW (contained in SWNLIB and listed in this report) or in the programme called [PENDLETN]FTSTMPFIT.

WAVENUMBER CONVERSION ROUTINES

```
      REAL FUNCTION GETSWN(W,COEF1,COEF2,COEF3,COEF4)
C TAKES THE REAL WAVENUMBER W AND CONVERTS IT TO SLIDE WAVENUMBER
C  EG. COEF1=0.1060792
C  EG. COEF2=-45204.29
C  EG. COEF3=-5.2041949E+10
C  EG. COEF4=4.1629131E-12
      REAL W
      REAL WW
      WW=W*W
      GETSWN=W*(COEF1+(COEF2+COEF3/WW)/WW+COEF4*WW)
      RETURN
      END
C
C
      REAL FUNCTION SWPRW(W,COEF1,COEF2,COEF3,COEF4)
C TAKES THE REAL WAVENUMBER W AND COMPUTES D(SIGMA)/D(NU) AT W.
C  EG.  COEF1=0.1060792
C  EG.  COEF2=-45204.29
C  EG.  COEF3=-5.2041949E+10
C  EG.  COEF4=4.1629131E-12
      REAL W
      REAL WW
      WW=W*W
      SWPRW=COEF1-(COEF2+3.*COEF3/WW)/WW+3.*COEF4*WW
      RETURN
      END
C
C
      REAL FUNCTION GETRWN(BIN,COEF1,COEF2,COEF3,COEF4)
C TAKES THE SLIDE WAVENUMBER BIN AND COMPUTES THE REAL WAVENUMBER W.
C  EG.  COEF1=0.1060792
C  EG.  COEF2=-45204.29
C  EG.  COEF3=-5.2041949E+10
C  EG.  COEF4=4.1629131E-12
      REAL BIN,W,CORR,GETSWN,SWPRW
      NTIME=0
      W=BIN/COEF1
100   CORR=(GETSWN(W,COEF1,COEF2,COEF3,COEF4)-BIN)
     1              /SWPRW(W,COEF1,COEF2,COEF3,COEF4)
      W=W-CORR
      NTIME=NTIME+1
      IF(NTIME.GT.1000)GO TO 90
      IF(ABS(CORR).GT.1.E-6*ABS(W))GO TO 100
      GETRWN=W
      RETURN
90    WRITE(*,*),  ' ERROR, COULD NOT CONVERGE'
      WRITE(*,*),' IN ',NTIME,' ITERATIONS'
      RETURN
      END
```